###############################################################################
# Базовый класс для всех модулей
###############################################################################
@CLASS
module
###############################################################################



###############################################################################
# метод выводит только базовые свойства страницы
# переопределяется в классах-наследниках
@content[]
<?xml version="1.0" encoding="utf-8" ?>
<document lang="ru">
	<head>
		<title>$_val.title</title>
		<description>$_val.description</description>
		<keywords>$_val.keywords</keywords>
	</head>
	<content><![CDATA[^untaint{$_val.content}]]></content>
</document>
###############################################################################



###############################################################################
# параметры XSLT-преобразования не заданы
@XMLparam[]
$result[^hash::create[]]
###############################################################################



###############################################################################
# список потомков
@menuOffsprings[id][off;link]
$off[^_core.getOffsprings[$id]]
$cur_page_uri[^_core.cropURI[$request:uri]/]
<offsprings id="$id">
  ^off.menu{
  	$link[^_core.id2uri[$off.id]]
  	<item
  	  ^if(^off.main.bool[]){main='true'}
  	  ^if($cur_page_uri eq $link){current='true'}
  	  href='$link' title='$off.title' id='$off.id'>$off.name</item>
  }
</offsprings>
###############################################################################



###############################################################################
# список соседних разделов
@menuBrothers[id][off;curObject;link]
$curObject[^_core.getObject[$id]]
$off[^_core.getOffsprings[$curObject.pid]]
$cur_page_uri[^_core.cropURI[$request:uri]/]
<core id="$id">
  ^off.menu{
  	$link[^_core.id2uri[$off.id]]
    <item
      ^if(^off.main.bool[]){main='true'}
			^if($cur_page_uri eq $link){current='true'}
      href='$link' title='$off.title' id='$off.id'>$off.name</item>
  }
</core>
###############################################################################



###############################################################################
# хлебные крошки
@breadCrumb[id][bc]
$bc[^_core.getBreadCrumb[$id]]
$cur_page_uri[^_core.cropURI[$request:uri]/]
<breadcrumb id="$id">
  ^bc.menu{
    <item
      ^if(^bc.main.bool[]){main='true'}
      ^if($cur_page_uri eq $bc.link){current='true'}
      href='$bc.link' title='$bc.title' id='$bc.id'>$bc.name</item>
  }
</breadcrumb>
###############################################################################